# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Arch Haskell Team <arch-haskell@haskell.org>

_hkgname=http-conduit
pkgname=haskell-http-conduit
pkgver=2.3.8.3
pkgrel=12
pkgdesc="HTTP client package with conduit interface and HTTPS support"
url="https://www.yesodweb.com/book/http-conduit"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-aeson' 'haskell-attoparsec' 'haskell-attoparsec-aeson'
         'haskell-conduit' 'haskell-conduit-extra' 'haskell-http-client' 'haskell-http-client-tls'
         'haskell-http-types' 'haskell-resourcet' 'haskell-unliftio-core')
makedepends=('ghc' 'haskell-blaze-builder' 'haskell-case-insensitive' 'haskell-cookie'
             'haskell-crypton-connection' 'haskell-data-default-class' 'haskell-hunit'
             'haskell-hspec' 'haskell-lifted-base' 'haskell-network' 'haskell-streaming-commons'
             'haskell-temporary' 'haskell-unliftio' 'haskell-utf8-string' 'haskell-wai'
             'haskell-wai-conduit' 'haskell-warp' 'haskell-warp-tls')
source=("https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz")
sha512sums=('9b665679b1baa5e487dbd786443317534641421ab90c6a0b16874b9bfdb3077c6e84b8c2afe063b7a356a79be2647ae4b4256a331f02da348664246734d708b5')

build() {
    cd $_hkgname-$pkgver

    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver

    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
